
======================CHUA_CHAOS===================================

MULTI-DIMENSION OSCILLATION WITH RANDOM CROSS-TALK
PLOTTED IN 2D.

�
CHUA_CHAOS
*
*   OUT1                                       1.3k +/-      OUT2
*  ____________________________________________/\  /\  /\______
* |      R1       | Chua's Circuit           |   \/  \/   |    |
* |_/\  /\  /\_   |                 +9V     _|_     R8   _|_   |
* |   \/  \/   |  |                    ^    ___C1      C2___   |
* |      290   |  |    VPL  47k  R4   /_\    |____________|    |
* |   |\       |  |      _/\  /\  /\___|   4.6nF  _|_   69n    |
* |___|+\ OUT3 |  |D1   |   \/  \/         +/-    ///          |
*     |  \_____|  |     |           3.3K             R9        |
* INN |  /     |  |_|\|_|_/\  /\R5/\____      _/\  /\  /\______|
* ____|-/      |  | |/|     \/  \/   |  |    |   \/  \/  65
* |   |/ OPA   |  |      _/\  /\  /\_| _|_   |  _  _  _
* |   R2       |  | D2  |R6 \/  \/3.3k ///   | / \/ \/ \ 8.5m
* |_/\  /\  /\_|  |     |                    |_| () () |_
* |   \/  \/290   |_|/|_|_/\  /\R7/\_                   _|_
* |                 |\|     \/  \/  _|_      VL         ///
* |_/\R3/\  /\_        VNG  47k     \ /
* 1.2k\/  \/  _|_                    V -9v
*             ///
*
*.OPTIONS  GMIN=1p      METHOD=TRAP   ABSTOL=1u     TEMP=27 srcsteps = 1  gminsteps = 1 
*.OPTIONS  RELTOL=.001  ABSTOL=1n     VNTOL=1n      ITL4=500     ITL1=400  
*======== ====== ====== ====== ====== ====== ====== ====== ====== 
VCC       VCC    0      DC     9
VEE       VEE    0      DC     -9

BOPA      OUT4   0      V =    9*tanh(tanh(tanh( (V(OUT1)-V(INN))*30)*30)*30)
ROP       OUT4   OUT5   5k
COP       OUT5   0      30p
BOUT      OUT3   0      V =   V(OUT5)

R1        OUT1   OUT3   290
R2        INN    OUT3   290
R3        INN    0      1.2K
R4        VPL    VCC    47K
R5        VPL    0      3.3K
R6        VNG    0      3.3K
R7        VNG    VEE    47K
R8        OUT1   OUT2   1.34045K
R9        OUT2   VL     65
C1        OUT1   0      4.6n  IC = 1
C2        OUT2   0      69n
L1        VL     0      8.5m
D1        OUT1   VPL    DD
D2        VNG    OUT1   DD

*=========The_Model_Files==================================
.model    DD  D( IS=3.15e-18 rs = 1m)

.tran      1u    15m    0       1u UIC

.control 
run
plot     out1  out2
plot     out1 vs out2
.endc   
.end

